

# 基於 HT95R5x/6x 之 FSK Decoder 功能之使用方法

文件編碼：HA0228T

## 簡介

CID (Calling Identity Delivery)，電話主叫識別訊息發送，也稱為來電顯示，是電信局向被呼叫的電話終端提供的一種的服務。CID 將主叫的電話號碼、用戶姓名和呼叫時間等資訊發給被呼叫端，被呼叫端接收並儲存這些資訊，以使用戶查閱。

來電顯示實現的一般方法是，發送端的程式控制交換機將主叫的電話號碼等相關資訊通過局間指令系統傳遞給終端交換機，終端交換機再將主叫訊息以 FSK (Frequency-shift Keying) 或 DTMF (Dual Tone Multi-Frequency) 方式，在第一次和第二次振鈴間隔期前傳送給被呼叫用戶終端設備。中國的通信行業，採用 FSK 方式提供來電顯示服務，當然被呼叫用戶之前須申請這項服務。

盛群半導體推出的 CID Phone 8-Bit MCU -- HT95R5x/HT95R6x，內置 FSK Decoder，本文將介紹如何使用 HT95R5x/HT95R6x 及其 FSK Decoder 實現來電顯示功能。

## 工作原理

CID 有 Bell 202 FSK 和 ETSI V.23 FSK 等多種規格。其中 Bell 202 是由 Bell 實驗室所制訂的 FSK CID 規範，主要在中國、新加坡和美國等國家使用，本文介紹的是 Bell 202 FSK 的規格。程式控制交換機傳輸 CID 資料的參數如下所示：


調製類型：BFSK (Binary Frequency-Shift Keying)

Mark (Logic 1) :  $1200 \pm 12\text{Hz}$

Space (Logic 0) :  $2200 \pm 22\text{Hz}$

傳輸速度：  $1200 \pm 12\text{bps}$

CID 資料的傳輸可分為兩類，一類是在電話機 On-hook (掛機) 狀態下傳輸，另一類是在電話機 Off-hook (摘機) 狀態下傳輸。On-hook 狀態的 CID 資料傳輸又可分為兩種，一種是聯合振鈴的，另一種是不聯合振鈴的。下圖是 On-hook 狀態下聯合振鈴的 CID 資料傳輸時序。




FSK Data 包含 Channel Seizure Signal、Mark Signal、Message 和 Checksum。


- Channel Seizure Signal  
這是發送 CID 資訊時首先要發送的信號，由一組 300 個連續的 "0" 和 "1" 交替組成。其第一位為 "0"，最後一位為 "1"。
- Mark Signal  
由 180 個 "1" 組成。
- Message  
Message 的格式有兩種，分別是 SDMF (Single Data Message Format) 和 MDMF (Multiple Data Message Format)。程式控制交換機向用戶端傳送的資訊是 8-bit 的 ASCII 格式的資料。
- Checksum  
一個位元組的資料，Message 的所有資料與 Checksum 相加，結果的最後一個位元組為 0。

SDMF 和 MDMF 資料格式如下圖所示。

**SDMF message**



**MDMF message**



下面是一組 FSK SDMF 資料：


04H，0FH，30H，31H，32H，33H，31H，36H，35H，39H，35H，36H，33H，31H，39H，39H，39H，39H，D8H。

上述資料中：

- 04H 表示 Message Type，是 SDMF
- 0FH 表示 Message Length
- 30H、31H、32H、33H 表示日期 01 月 23 號
- 31H、36H、35H、39H 表示時間 16：59
- 35H、36H、33H、31H、39H、39H、39H、39H 表示電話號碼 5631999
- D8H 是 Checksum，用於校驗 CID 資料

HT95R5x/HT95R6x 內置 FSK decoder，支持 Bell 202 和 ETSI V.23 兩種 CID 的規格，具有 Ring 和 Line Reversal 的偵測功能。FSK Decoder 支持四種能夠產生外設中斷的中斷源，分別是 FSK RAW Data 的下降緣、Ring 或 Line Reversal 偵測、FSK 載波偵測和 FSK 資料封包。





FSKDF：FSK RAW 資料下降緣引起的中斷旗標，中斷發生時，設定為 1。

DTMFF：DTMF 接收器中斷旗標。

EFSKDI：FSK RAW 資料下降緣引起中斷的使能，1 為使能，0 為除能。

EDTMFI：DTMF 接收器中斷使能，1 為使能，0 為除能。


需要注意的是，EFSKDI 使能時，控制器會自動將 RMSK、CMSK 和 FMSK 除能。如果中斷允許，FSKDF、RDETF、CDETF 和 FSKF 都會引起中斷入口地址為 10H 的外設中斷，且退出中斷服務副程式時，不會被硬體清除為 0，需用軟體清除為 0，否則不能進入下一次的中斷。

HT95R5x/6x 的 FSK Decoder 模組解析電話線 TIP 和 RING 上的 FSK 信號，並且可產生兩種格式的資料，分別是串列資料和 8-bit Packet 資料。為了 FSK Decoder 能夠正常工作，需將暫存器 FSKC 中的 F\_PWDN 位清除為 0。串列的 FSK 資料用兩種格式表示，分別是 RAW Data 和 COOK Data，並且分別可用旗標 DOUT 和 DOUTC 監視。當 Decoder 處於工作模式時，旗標 DOUT 指示 Decoder 的輸出，輸出的資料流包括 Channel Seizure Signal (由 0 和 1 交替組成) 和 Message；旗標 DOUTC 指示 Decoder 的輸出與 DOUT 的相似，但不包括 Channel Seizure Signal。若 FSK 資訊沒有被偵測到，則 DOUT 和 DOUTC 都保持高準位。因此用戶可用旗標 DOUT 和一個定時器用軟體實現的 FSK 解碼功能。


除了串列資料的方式外，Decoder 也提供 FSK Packet 資料。當 Decoder 接收到一個 FSK 信號時，Decoder 把 10-bit 的資料打成一個 8-bit 的 packet 資料，10-bit 資料中的第一位和第十位都被忽略。當 8-bit Packet 資料有效時，會被儲存到 FSK 資料暫存器 FSKD 中，並且 FSK Packet 資料中斷旗標 FSKF 會被設定為 1。如果 FSKC 暫存器的 FMSK 為 0，且外設中斷得到允許，這會引起一個週邊中斷。FSK Decoder 可以將 DOUT 的資料封包，也可將 DOUTC 的資料封包，這由 FSKC 的 FSKSEL 位元決定。需要注意的是能夠被封包的 10-bit 資料的第一位必須為 0，因而 Mark Signal 的資料不會被封包。

為了能夠偵測 FSK 載波信號、解析出串列資料和將 10-bit 資料轉成一個 8-bit 的資料包，控制器的必須工作在 Normal 模式。當工作在 Green 和 Sleep 模式時，FSK Decoder 將解出錯誤的資料，但 Ring 和 Line Reversal 的偵測功能仍然有效。

FSK Decoder 時序如下圖所示。




### 應用電路圖



(參考 HT95R5x data sheet)

## S/W 流程圖



## 程式說明

程式設定在 On-hook 狀態下接收 SDMF 和 MDMF 格式的 CID 資料，並處理、保存在 RAM Bank1 中。SDMF 的資料包括時間（如 03 月 17 日，16 時 18 分）和電話號碼（如 5551212）；MDMF 的資料包括時間、電話號碼和姓名（如 John Smith）。

## 程式範例

```

; Option: WDT Disable
include HT95R55.inc
ds .section 'data'
VAR0 DB ?
RX DB 40DUP(0) ;save CID information
COUNT0 DB ?
COUNT1 DB ?
COUNT2 DB ?
LENGTH DB ? ; CID Length from CID
LENGTH1 DB ? ; LENGTH1=LENGTH+3, (Compared with LENGTH, Add message type,
; message length and checksum)
FLAG_S DBIT ; Receive Single Data Message Format Flag
FLAG1 DBIT ; Finish receiving flag
FLAG0 DBIT ; Receive channel seizure signal 55H
FLAG_M DBIT ; Receive Multiple Data Message Format Flag
cs .section 'code'

ORG 0000H
JMP MAIN_START
ORG 0004H
    
```

```

RETI
ORG 0008H
RETI
ORG 000CH
RETI
ORG 0010H
JMP FSK_INT
ORG 0014H
RETI
ORG 0018H
RETI

ORG 0020H

MAIN_START:
SET UPEN
CALL DELAY_20MS
SET MODE1
CLR F0
CLR F1 ; System Clock: HCLK 3.58MHz
CALL CLEAR_RAM0
MOV A, 01H
MOV BP, A
CALL CLEAR_RAM1 ; Clear RAM bank0 and bank1
MOV A, OFFSET RX
MOV MP0, A
MOV A, 40H
MOV MP1, A

CLR F_PWDN ; FSK decoder work at operation mode
CLR CMSK ; enable Carrier detect interrupt
CLR RMSK ; enable Ring or line reversal detect interrupt
CLR FMSK ; enable FSK packet data interrupt
CLR EFSKDI ; disable FSK RAW data falling edge interrupt
CLR FSKSEL ; select FSK packet data source DOUT
SET EMI
SET EPERI ; enable peripheral interrupt

L0:
SZ FLAG_S
JMP L1
SZ FLAG_M
JMP L1
JMP L0

L1:
SNZ FLAG1
JMP L0
CLR EPERI
MOV A, 01H
MOV BP, A
MOV A, OFFSET RX
MOV MP0, A

L2:
MOV A, IAR0
MOV IAR1, A ; store the CID from RX to ram bank1
INC MP0
INC MP1
INC COUNT0
MOV A, LENGTH1
SUB A, COUNT0
SNZ C
JMP L3

```

```

MOV A, COUNT0
XOR A, LENGTH1
SNZ Z
JMP L2
INC COUNT2
L3:
CLR FLAG_S
CLR FLAG_M
CLR FLAG1
CLR FLAG0
CLR COUNT0
CLR COUNT1
CLR LENGTH
CLR LENGTH1
MOV A, OFFSET RX
MOV MP0, A
SET EPERI
JMP L0

FSK_INT:
CLR EPERI
SZ CDEF          ; check FSK carrier detect interrupt flag
CLR CDEF
SZ RDEF          ; check ring or reversal detect interrupt flag
CLR RDEF
SNZ FSKF         ; check FSK packet data interrupt flag
JMP EXIT_INT
CLR FSKF
SZ FLAG1
JMP EXIT_INT
CALL DECODE_SUB

EXIT_INT:
CLR PERF
CLR FSKDF
CLR RDEF
CLR CDEF
CLR FSKF
SET EPERI
RETI

DECODE_SUB:
SZ FLAG0
JMP SUB0
MOV A, FSKD
XOR A, 55H
SNZ Z
RET
SET FLAG0          ; detect FSK channel seizure signal 55H
SUB0:
SZ FLAG_S
JMP M1
SZ FLAG_M
JMP M1
MOV A, FSKD
XOR A, 04H         ; check SDMF message type
SZ Z
JMP S0
MOV A, FSKD
XOR A, 80H         ; check MDMF message type
SZ Z

```

```

        JMP M0
        RET
S0:
        SET FLAG_S
        JMP M1

M0:
        SET FLAG_M

M1:
        INC COUNT1
        MOV A, COUNT1
        XOR A, 02H
        SNZ Z
        JMP M2
        MOV A, FSKD
        MOV LENGTH, A        ; get message length
        ADD A, 03H
        MOV LENGTH1, A

M2:
        MOV A, FSKD
        MOV IAR0, A        ; save the CID information to RX
        INC MP0
        MOV A, COUNT1
        SUB A, 3
        SNZ C
        RET
        MOV A, COUNT1
        XOR A, LENGTH1
        SZ Z
        SET FLAG1        ; check if Receive action finish
        RET

CLEAR_RAM0:
        MOV A, 40H
        MOV MP0, A

CLEAR0:
        CLR IAR0
        MOV A, MP0
        XOR A, 0FFH
        SZ Z
        RET
        INC MP0
        JMP CLEAR0

CLEAR_RAM1:
        MOV A, 40H
        MOV MP1, A

CLEAR1:
        CLR IAR1
        MOV A, MP1
        XOR A, 0FFH
        SZ Z
        RET
        INC MP1
        JMP CLEAR1

DELAY_20MS:
        MOV A, 165
        MOV VAR0, A
    
```

```
DELAY:  
SDZ VAR0  
JMP DELAY  
RET
```

## 結論

本文範例程式可以實現 SDMF 和 MDMF 的 CID 資料解碼，用戶可參考使用。